

マイクロコンピュータによる TSS 端末プログラムの開発

黒木 莊一郎
(受理 昭和60年5月31日)

DEVELOPMENT OF A TSS TERMINAL CONTROL PROGRAM FOR A MICRO COMPUTER

Souichirou KUROKI

The communication program between a host computer and a micro computer is developed in this paper. The program is written in N88-DISK BASIC (86) and the micro computer is a PC-9801 (NEC).

This program enables file transmission between the host (MELCOM COSMO 800 III & FACOM M-382) and the terminal. The file transmission procedure is described in detail.

The program listing and the manual are included in the appendix.

§1 はじめに

マイクロコンピュータは近年の急激な性能の向上、価格の低下によってかなり普及してきている。それとともに周辺機器（プリンタ・XYプロッタ・A/Dコンバータ等）も比較的手軽に使用できるようになった。また大学等の HOST コンピュータも TSS 処理が主流となり、鹿児島大学においても昭和57年度の計算機更新時から TSS 処理になっている。同じ頃16ビットのマイクロコンピュータが販売されはじめ、大容量・高速処理が研究室単位でパーソナルに利用されている。

16ビットマイクロコンピュータが大容量・高速処理と言っても、8ビットマイクロコンピュータに比較してのことであり、容量・速度の面で大型コンピュータに及ぶくもない。マイクロコンピュータはパーソナルな利用に適しており、処理速度を問題としない科学技術計算や、実験等の計測制御に今後ますます利用されていくものと考えられる。しかし、マイクロコンピュータの処理速度及び使用可能な周辺機器には限界があり、また大型コンピュータには不特定多数のユーザが限られた周辺機器を利用するため小回りが効かないという不便さもある。そこでマイクロコンピュータを

大型コンピュータの TSS 端末化し、さらにプログラムやデータの転送を可能にすることによって、お互いの長所を生かした利用をすることができ便利である。さらに他大学の大型コンピュータにも N1 ネットワークで接続されており、研究室から他大学の HOST コンピュータにアクセスすることが可能であり、より高速な計算処理やデータベースが簡単に利用できる。

マイクロコンピュータは通信機能を標準で装備しており、BASIC 言語で通信制御プログラムを記述できる。従って HOST コンピュータの通信手続きにのっとった処理をマイクロコンピュータ側で行えば比較的簡単にプログラムを作成できる。またプログラムやデータ等のファイル転送については HOST コンピュータの機種によってサポートの有無があり、転送手順もいくつかのレベルが規格化されており複雑である。しかし、HOST コンピュータの適当なコマンドをシミュレートすることによりリファイル転送機能を実現することができる。

本報告では鹿児島大学の HOST コンピュータ (MELCOM COSMO 800 III) 及び N1 ネットワークによって接続できる九州大学の HOST コンピュータ (FACOM M-382) を対象としたマイクロコンピ

ユータの TSS 端末プログラムについて述べる。使用機種は NEC PC-9801 であり、使用言語は N88-DISK BASIC (86) とした。

§2 構成機器と通信手続き

構成機器は図-1 に示すとおりである。MELCOM




図-1 構成機器

COSMO 800 III の TSS 端末としてマイクロコンピュータの具備すべきハードウェア仕様としては (1) RS232C インターフェース, (2) 4 線式 (全二重), (3) 無手順, (4) 回線速度は音響カプラの場合 300bps, モデムの場合 1200bps である。またデータ・コード形式は 7 ビット (ASCII) + 1 ビット (パリティ・偶数) + 1 ビット (スタート) + 1 ビット (ストップ) となっている。PC-9801 の場合、ハードウェアの設定としては、(4) の回線速度に合わせてメモリスイッチを書き換える程度の操作ですむ。一旦書き換えておけば電源を切っても残っており、以後いちいち設定する必要はない。データ・コード形式は BASIC プログラム内で自由に設定できる。N1 ネットワークによって他大学の HOST コンピュータに接続する場合にも MELCOM の NVT サーバの管理下に置かれるので、特に変更は必要ない。次章からは上記の条件をもとにしたプログラムの開発について述べる。

§3 通信プログラム

マイクロコンピュータで通信を行うためには、一般

の端末機をエミュレートするプログラムを作成しなければならない。その一つの手順としては (1) 回線に HOST コンピュータからの受信データがあるか, (2) あればそのデータをディスプレイ上に表示する。 (3) キーボードからの入力データがあるかどうか, (4) あればそのデータを HOST コンピュータに送信する。そして (1) に戻るといふ繰り返しである。その際、通信は無手順で行われるため HOST コンピュータからのデータを取りこぼさないために受信処理を優先する必要がある。通信プログラムのフローチャートを図-2 に示す。この他通信回線の割り込みを利用する方式




図-2 通信のフローチャート

```

140 OPEN "COM1:E71NS" AS #1 : PRINT #1,
150 L=LOC(1) : IF L THEN 180
160 A$=INKEY$ : IF A$<>" " THEN PRINT #1,A$;
170 GOTO 150
180 PRINT INPUT$(L,1); : GOTO 150
    
```

図-3 通信プログラムの例

もあるが、今回は図-2 のようなソフトウェアスキャン方式を採用している。図-3 にソフトウェアスキャン方式による通信プログラムの例をあげる。図のようにわずかな行でプログラムを組むことができるが、これに各種の便利な機能を組み込んだり、TSS 端末動作を正常に行わせるためにはいろいろなルーチンが必要となる。このように開発したプログラムの全リストを付録に示している。次章からは各機能についてプログラムと対応させて述べる。

§4 機能及びプログラムの説明

主な機能は TSS 端末としての動作と HOST コンピュータとのファイル転送機能に集約できる。プログラムは送受信の文字データ処理が主体となる。

N88-DISKBASIC (86) では文字データ処理を多用するとガベージコレクション (文字列変数のちり集め) によってある間隔毎 (最大約 60K バイトの文字処理) にプログラムの実行が一次停止するという不都合な問題がある。特にファイル受信時にガベージコレクションが起きると受信バッファのオーバーフローを生じ、それ以後のデータは取り込むことができない。従ってファイル受信時にはガベージコレクションを強制的に起こさせる FRE(0) 関数を使いソフト上の対策を行っている。また、通常の交信を長く続けているとやはりガベージコレクションが生じ、それ以降交信が続行できなくなる。この対応をプログラム利用者に意識させずに行うことは N88-DISKBASIC (86) でプログラムを作成する限り不可能に近い。(機械語では可能) 一方 PC-9800 は強力な TERM (ターミナル) コマンドをもっており、このコマンドを実行した状態のターミナルモードで交信する限りガベージコレクションは生じない。従って本プログラムではプログラムで交信する BASIC モードからターミナルモードへ、またターミナルモードから BASIC モードへと切り換えることによって対応している。

§4. 1 初期設定

プログラムの 90 ~ 190 行で初期設定を行っている。配列 MPS はファイル受信時のバッファとして使う。120~130 行は主に制御コードの文字変数を設定している。CSRCL\$ はカーソルを消去するための文字列である。CTL\$ は制御キー入力をメインループで判別するための文字列であり、受け付ける制御キー (アスキーコード 0~31 まで) としては改行キー・水平タブキー・エスケープキー・バックスペースキーの 4 つである。140 行以降はタイトル表示、プリンタの印字モード設定、ディスプレイの表示設定、ファンクションキーの設定等である。

§4. 2 メインループ

200~380 行が HOST コンピュータとの通信処理のメインループとなっている。基本的には §3 で述べたソフトウェアスキャン方式のプログラムである。追加された機能としては HELP キー割り込みによるファンクションキー説明表示、STOP キー割り込みによるブレイク信号送出、ファンクションキー割り込みによる各種機能、カーソル表示、プリンタへの交信のエコーバック及びターミナルモードとの切り換えであ

る。

図-3 のプログラムで通信する場合、カーソルが表示されずに不安である。そこで BASIC によるカーソル表示機能を付加した。カーソル表示は 280・285 行で行っている。まず現在のカーソル位置 (X 桁, Y 行) を調べ、その位置にカーソルを表示する。カーソルの形はユーザの好みで文字コードの中から一文字選択できる。本プログラムではターミナルモードのカーソルと区別するため “アンダーバー” を使用している。次のその位置がスクロール画面の最下行で 79 桁目 (一行の最後の桁) であれば自動的にスクロールされるのでカーソル行を一行減じる。そしてカーソル文字を点滅させ、カーソル位置をずらす。以上の手順でカーソル表示機能を実現している。通信回線にデータが受信されたり、他の機能の処理を行う場合はカーソルを消去せねばならない。その場合はカーソル消去文字列 (CSRCL\$) を使って消し各処理を実行するようにしている。そのため若干プログラムが複雑になってしまった。

320 行では制御キー入力の判別を行っている。受け付ける制御キーは前述したように 4 種類である。ただしバックスペースキーは文字消去に使用すると不都合が生じるので、デリートキーコード (127) に変更して送信するようにしている。370 行では交信をプリンタへエコーバックする。ただしプリンタの印字速度によっては正常に印字できない場合があるので、注意を要する。通信速度を 300bps (音響カプラ使用時) で行う場合は問題ないが、1200bps (モデム使用時) で行う場合、プリンタのバッファ容量との兼ね合いで決まるので、改行時間も含めて毎秒 120 文字以上の印字速度をもつプリンタが必要である。ファンクションキーによる各種機能については付録を参照されたい。

§4. 2 ファイル転送

本プログラムの主目的はこのファイル転送機能を実現することにあった。ファイル転送のための通信ルーチンは図-2, 3 と基本的には同じであり、ディスプレイ表示及びキーボード入力がそれぞれディスクへの書き出し及び読み込みになる。ただし 1200bps の通信速度に追いつくために、1 行単位 (改行コードまでの文字列) の処理を行わねばならない。また HOST コンピュータのファイル転送に利用できる適当なコマンドを選択し、その手順を調べなくてはならない。ファイル転送を実現するためのフローチャートを図-4




図-4 ファイル送信のフローチャート

(ファイル送信). 図-5 (ファイル受信) に示す。図-5でわかるようにファイル受信時のガベージコレクション対策が必要であり、最大約 60K バイトの文字列を受信すると起きるようである。1 行の文字数をラインプリンタイメージの 136 桁とすると約 440 行分に相当する。プログラムでは 400 行受信する毎に一旦ディスクへ書き出し、配列を消去した後ガベージコレクションを FRE (0) 関数で強制的に起こし、また配列を宣言するという処理で対応している。400 行という単位は PC-9800 のユーザ RAM 容量で文字領域が最大に取れる場合であって、RAM を拡張していない場合は、400 行より少なくする必要がある。これはプログラムの 100 行にある変数 MPMX の値を変更すればよい。(100 にしておけば安全)

690 行からのファイル転送ルーチンはファンクションキーの 5 をたたくことによって起動する。ファイル転送は鹿児島大学の MELCOM と九州大学の FACOM の 2 つの HOST コンピュータに対してサポートしている。プロンプトが異なることを利用してどちらの HOST コンピュータと接続されているか 700~830 行で判別し、機種に応じたファイル転送ルーチンへ飛んでいく。MELCOM とのファイル転送ルーチンは 840 行から初まっており、受信か送信かを聞いてくる。受信の場合は 940~1260 行で処理を行っている。ファイル受信に利用する MELCOM のコマンドは COPY [転送したいファイル名] TO ME コマンドである。ガベージコレクションは 1050 行、1210 行で強制的に起こさせている。送信の処理は 1270~1630 行で行っており、先ず MELCOM で




図-5 ファイル受信のフローチャート

作られるファイル名を入力するとそのファイルを消去していることに注意されたい。次に COPY ME TO [作成されるファイル名] コマンドによってファイル送信を開始する。一行送信する毎に MELCOM からはプロンプト (ピリオド) が送られてくるのでそれを待って次の一行を送信するという図-4 に示した手順で処理している。送信を完了すると行番号付きのファイルへ変換する作業を行い送信ルーチンから抜け出す。この作業によって MELCOM 内に作成されたファイルはエディタで表示でき修正することが可能となる。

FACOM とのファイル転送ルーチンは 2470 行から初まり、ファイル受信処理は 2560~2920 行で行う。手順は MELCOM の場合と同様であり、LIST [転送したいファイル名] コマンドを利用している。ファイル送信には EDIT [送信したいファイル名] NEW

〔ファイル識別子〕コマンドの INPUT モードを利用しており、2930～3310 行で行っている。ファイル転送の交信が成立せずエラーメッセージが送られてきたときの処理がそれぞれ含まれている。なおファイル転送異常時には STOP キー割り込みによってただちにディスクファイルはクローズされ通常の交信ルーチンに戻ることにしており、それ以後のファイル転送されたデータはディスクに書き込んでいない。

§5 おわりに

本プログラムの主目的は HOST コンピュータとマイクロコンピュータ間のファイル転送機能を実現することである。これによって科学技術計算や計測データ処理等のコンピュータ活用の一助となると考えている。プログラミングに際しては解り易いメッセージを表示し、操作ミスを防ぐようにしているが、好みにあわせて適宜プログラムの変更をされても構わない。

今後、スクリーンエディタがマイコン上で利用でき

るよう本プログラムを拡張していく予定である。

なお本報告は第 19 回情報処理研究会で発表したものの一部である。

参考文献及び資料

- 1) PC-9801 BASIC リファレンスマニュアル NEC
- 2) PC-9801 ユーザーズマニュアル NEC
- 3) MELCOM UTS/VS TSS 入門書 三菱電機(株)
- 4) MELCOM UTS/VS TSS 使用手引書 三菱電機(株)
- 5) 入江他：FM-8 用インテリジェント端末プログラム 九州大学大型計算機センター広報 Vol.15 №5 1982
- 6) N1 ネットワークの利用法 技術解説 三菱電機(株)、鹿児島大学電子計算機室 昭和59年9月
- 7) 鹿大端末による九大大型計算機の利用 鹿児島大学電子計算機室講習会テキスト 昭和59年

付 録

鹿大 T S S 端末プログラム (KTCP) VER. 2.19 使用法

1. プログラムの実行

プログラムを走らせると先ずプリンタ使用の有無をきいてくる。

1.1 プリンタ設定

メッセージ	キーイン	備 考
> use PRINTER (Y/N) ?	プリンタを使わなければリターンキー プリンタを使う場合はYで次へ	1. 2 へ
>> PRINTER ON !! (PRINTER MODE MENU)		プリンタの電源をいれる。 プリンタのマニュアル参照
>>> SELECT PRINTER MODE ?	N~Hから選択	1. 2 へ

・プリンタ使用上の注意：①プリンタはNEC製を対象としているため他社製を使用する場合は1980~2000行、570行の変更を必要とする場合がある。

②ファイル転送以外の交信ではプリンタへのエコーバックをサポートしているが、転送速度(300または1200ボー)とプリンタの印字速度・バッファ容量とのかねあいで印字が正常に出来ない場合がある。

1.2 LOGON

オープニングメッセージ MELCOMからの オープニングメッセージ LOGON PLEASE:	(STOP) 登録番号等	BREAK信号送出。モデムの場合。 音響カブラの場合は無視。 * f 4 キー
--	---------------------	---

以下PC-9800はTSS端末として動作する。

・f 4 キーによる登録番号等の自動送信：f 4 キーを押すとプログラムの2300~2370行で設定された登録番号・ユーザ名・パスワードが送信できる。

2. ファンクションキーの機能(プログラムモード)

- [f 1] ・ ・ ターミナルモードに入る。(shift)+(STOP)でプログラムモードに戻る。
- [f 2] ・ ・ キー入力中の文字を全て抹消する。(キャンセル信号送出)
- [f 3] ・ ・ N 1 ネットワークへのリンクコマンド送出 [f 4] ・ ・ 登録番号等の自動送信(2.2を参照)
- [f 5] ・ ・ 受信の一時停止。(HALT信号送出) [f 6] ・ ・ ファイル転送モードに入る。(4.を参照)
- [f 7] ・ ・ フロッピィディスク内のディレクトリ表示。
- [f 8] ・ ・ プリンタのon/offスイッチ(2.1でプリンタの設定をしていなくても high speed pica mode で動作する。)
- [f 9] ・ ・ フロッピィディスク内のファイルを指定してその内容を表示。(SPACE) キー → 押している間表示スピードが遅くなる。(CTRL)+S → 一時停止
(2.1の注意にあるようにオンラインでのプリンタへのエコーバックが正常に行なわれない場合は一旦ファイル転送してきてこの機能とf 8 キーによってオフラインで印字できる。)
- [f 1 0] ・ ・ LOGOFFの後プログラムエンド、BASICモードにもどる。
- [HELP] ・ ・ ファンクションキーの機能説明表示。
- [STOP] ・ ・ BREAK信号送出またはファイル転送モード異常時の割り込み。

3. ファイル転送モード

f 6 キーをおすとファイル転送モードにはいる。f 6 キーをおす場合は原則として、鹿大コンピュータではTELモード（プロンプトが !）、九大コンピュータではコマンドモード（プロンプトが READY）とする。

> HOST SENSE ... と表示しHOSTコンピュータを識別し、各HOSTコンピュータに対応したメッセージが表示される。（鹿大のみのユーザは710行をGOTO 840に変更してもよい。）

3.1 鹿大コンピュータ (MEL.COM COSMO 800 III) とのファイル転送

転送方向の選択

メッセージ	キーイン	備考
> < KADAI > 1: MELCOM --> PC-9800		
> 2: PC-9800 --> MELCOM		
> SELECT (1 OR 2 :cancel -> other key) ?	1 (受信)、2 (送信)	

ファイル受信 (TELモードの COPY 'ファイル名' TO ME コマンドをシミュレート)

> MELCOM ---> PC-9800		
> MELCOM filename ?	ファイル名	
> PC-9800 drive NO. ?	ドライブ番号	リターンキーのみ→ドライブ1になる。
> PC-9800 filename ?	ファイル名	
> SURE (Y/N) ?	YまたはN	

ファイル送信 (TELモードの COPY ME TO 'ファイル名' コマンドをシミュレート)

> PC-9800 ---> MELCOM		
> PC-9800 drive NO. ?	ドライブ番号	リターンキーのみ→ドライブ1になる。
> PC-9800 filename ?	ファイル名	
> MELCOM filename ?	ファイル名	
> SURE (Y/N) ?	YまたはN	

3.2 九大コンピュータ (FACOM M-382) とのファイル転送

転送方向の選択

メッセージ	キーイン	備考
> * KYUDAI * 1: FACOM --> PC-9800		
> 2: PC-9800 --> FACOM		
> SELECT (1 OR 2 :cancel -> other key) ?	1 (受信)、2 (送信)	

ファイル受信 (コマンドモードの LIST データセット名、データセット識別子 コマンドをシミュレート)

> FACOM ---> PC-9800		
> FACOM 'DSNAME' ?	データセット名	
> ... QUALIFIER - ?	データセット識別子	FORT/FORT77/DATA/TEXT etc...
> PC-9800 drive NO. ?	ドライブ番号	リターンキーのみ→ドライブ1になる。
> PC-9800 filename ?	ファイル名	
> SURE (Y/N) ?	YまたはN	

ファイル送信 (コマンドモードの EDIT データセット名 NEW データセット識別子 コマンドによる
INPUTモードをシミュレート)

>	PC-9800 ---> FACOM		
>	PC-9800 drive NO. ?	ドライブ番号	リターンキーのみ→ドライブ1になる。
>	PC-9800 filename ?	ファイル名	
>	FACOM 'DSNAME' ?	データセット名	
>	... DATA SET TYPE - ?	データセット識別子	FORT77(FIXED)/DATA/TEXT etc...
>	SURE (Y/N) ?	YまたはN	

* ファイル転送等の異常時における処理: STOPキーによる割り込みでほしい復旧するが、それでもリターンキー等のキー入力を受けつけない状態になった時は伝家の宝刀? (全ての電源を切ってプログラムのロードからやり直し)を使用する。

(以下プログラムリスト)

```

10 'KADAI TSS TERMINAL CONTROL PROGRAM FOR PC-9800 'KTCP219'
20 '                                     Ver. 2.19      by S. Kuroki           :85/05/18
30 '      モノクロディスプレイ カーソル ヒョウシ             :83/06/22
40 '      STOP キー ニヨル BREAK シンゴウ                   :83/08/31
50 '      BIG FILE (>80モシ*500キョウ) ノ シュシソ ニオケル カヘーシ・コレクション ショリ :83/12/07
60 '      N1 network ニヨル キュウタイ(FACOM)トノ ファイルテンソウ :85/02/19
70 '      terminal mode <--> program mode ノ キリカエ       :85/02/22
80 '
90 DEFINT A-Z
100 MPMX=400:MPC=0:DIM MP$(MPMX)
110 C1=0:C2=0:C3=0:C4=0:C5=4:C6=6:C7=0:RSSW=0
120 CSRCL$=" "+CHR$(8):CR$=CHR$(13):LF$=CHR$(10)
130 ESC$=CHR$(&H1B):HALT$=ESC$+"H":CTL$=CR$+CHR$(9)+ESC$+CHR$(8)
140 CLS:GOSUB *TITLE
150 GOSUB *LPTSET
160 GOSUB *CRTSET
170 GOSUB *TITLE:GOSUB 2030
180 GOSUB *KEYSET
190 COLOR C6:PRINT "(MODEM):STOPキー ニテ BREAKシンゴウヲ オクotte クタサイ!!":COLOR C4
200 K=0:K2=0
210 OPEN "COM1:E71NS" AS #1: IF RSSW THEN PRINT#1,CHR$(&H18);CR$;:RSSW=0
220 ON HELP GOSUB 1650:HELP ON
230 B$="":ON STOP GOSUB 530:STOP ON
240 KEY ON
250 ON KEY GOSUB 430,510,520,2280,590,700,1680,560,1760,390
260 ON ERROR GOTO 480
270 '***** MAIN LOOP
280 X=POS(0):Y=CSRLIN:PRINT CHR$(&H80);:IF X=79 AND Y=24 THEN Y=23
285 COLOR@(X,Y)-(X,Y),2:LOCATE X,Y
290 L=LOC(1):IF L THEN PRINT CSRCL$;:GOTO 340
300 A$=INKEY$:IF A$="" THEN 290
310 IF A$<" " THEN 320 ELSE 330
320 LL=INSTR(CTL$,A$):IF LL=0 GOTO 290 ELSE IF LL=4 THEN A$=CHR$(127)
330 PRINT #1,A$;:GOTO 280
340 C$=""
350 C$=C$+INPUT$(L,#1):IF RIGHT$(C$,1)<>CR$ GOTO 370
360 L=LOC(1):IF L=0 THEN 360 ELSE 350
370 PRINT C$;:IF K2 THEN LPRINT C$;
380 L=LOC(1):IF L=0 THEN 280 ELSE 340
390 '=====
400 CLOSE:KEY OFF:HELP OFF
405 KEY 1,"load "+CHR$(&H22):KEY 2,"auto "
406 KEY 3,"go to ":KEY 4,"list ":KEY 5,"run"+CR$
410 CONSOLE 0,24,1,0:CLS:END:
'----- END

```

```

420 '=====
430 '----- TERMINAL MODE
440 ON ERROR GOTO 0:STOP OFF:HELP OFF:KEY OFF:CONSOLE 0,24,1:CLS
450 PRINT #1,:CLOSE:TERM "E71NS":GOSUB *CRTSET:RSSW=1
460 RETURN 200
470 '*****
480 RESUME 280
490 '*****
510 PRINT CSRCL$;:PRINT #1,CHR$(&H18);:RETURN 280 '----- CANCEL
520 PRINT CSRCL$;:PRINT #1,"NTSS ":K=1:RETURN 280 '----- NTSS
530 PRINT CSRCL$;
540 FOR I=1 TO 700:OUT &H32,&H3F:NEXT:OUT &H32,&H37:RETURN 280'----- BREAK
550 '----- LPT on/off
560 IF K2=1 THEN K2=0 :COLOR C6:PRINT "PRINTER OFF !":BEEP:COLOR C4:PRINT #1,:RE
TURN 280
570 IF LPT$="" THEN LPT$="N"
580 IF K2=0 THEN K2=1 :LPRINT ESC$;LPT$;:COLOR C6:PRINT "PRINTER ON !":BEEP:COLO
R C4:RETURN 280
590 PRINT CSRCL$;:PRINT #1,HALT$;:RETURN '----- HALT
600 *CRTSET
610 WIDTH 80,25:CONSOLE 0,25,0,0:COLOR C7:CLS:COLOR 4
620 PRINT "1:TERM 2:CAN 3:NTSS 4:ID 5:HALT ";
630 PRINT " 6:TRANS 7:FILES 8:LPTon/off 9:checkF 10:END "
640 COLOR C7:CONSOLE 1,25:COLOR C4
650 RETURN
660 *KEYSET
670 KEY 1,"PLATEN ":KEY 2,CHR$(&H18):KEY 3,"NTSS"+CR$:KEY 5,HALT$
680 RETURN
690 REM ***** SELECT HOST
700 PRINT CSRCL$;:KEY OFF: ON STOP GOSUB *INTRPT : STOP ON
710 GOSUB *HOSTSENSE:GOTO 740
720 PRINT:PRINT "> ** SELECT HOST **"
730 INPUT"> 1:KADAI(default) 2:KYUDAI ( 1 OR 2 ) ";KK
740 IF KK=0 OR KK=1 THEN 840
750 IF KK=2 THEN *KYUDAIF ELSE BEEP:GOTO 720
760 *HOSTSENSE
770 PRINT:PRINT "> HOST SENSE ... ":PRINT
780 PRINT #1,"";CR$;:B$=""
790 L=LOC(1):IF L=0 GOTO 790
800 B1$=INPUT$(1,1):IF B1$="!" THEN KK=1 :RETURN
810 B$=B$+B1$:IF INSTR(B$,"READY ")=0 THEN 790
820 IF INPUT$(1,1)<>LF$ THEN 820 ELSE KK=2
830 RETURN
840 '***** FILE TRANSFER FOR KADAI
850 ON ERROR GOTO *ERTRP
860 PRINT #1,HALT$;:C=0
870 PRINT "> < KADAI > 1: MELCOM --> PC-9800 "
880 PRINT "> 2: PC-9800 --> MELCOM "
890 INPUT "> SELECT ( 1 OR 2 :cancel -> other key ) ";C:PRINT
900 IF C<1 OR C>2 THEN PRINT #1,CR$;CR$;:GOTO 1250
910 INPUT "> SURE ( Y/N ) ";YN$:PRINT
920 IF YN$<>"Y" THEN 870
930 IF C=2 THEN GOTO 1280
940 '----- MELCOM --> PC
950 PRINT "> MELCOM ---> PC-9800 "
960 INPUT "> MELCOM filename ";F1$:PRINT
970 INPUT "> PC-9800 drive NO. ";F2$
980 IF F2$<>" " THEN F2$=F2$+" ":
990 INPUT "> PC-9800 filename ";F2N$:PRINT
1000 F2$=F2$+F2N$
1010 INPUT "> SURE ( Y/N ) ";YN$
1020 IF YN$<>"Y" THEN 870
1030 OPEN F2$ FOR OUTPUT AS #2
1040 PRINT#1,ESC$;ESC$;"QUIT";
1050 PRINT "> !!!!! WAIT A MINUTE !!!!!":MPF!=FRE(0)
1060 PRINT #1,CR$;
1070 PRINT #1, "C ";CHR$(&H27);F1$;CHR$(&H27)
1080 IF INPUT$(1,1)<>"C" THEN 1080

```

```

1090 IF INPUT$(1,1)<>LF$ THEN 1090
1100 N=1:LCNT=1
1110 MPC=1:NE=1
1120 LINE INPUT #1,MP$(MPC)
1130 IF LEN(MP$(MPC))=0 THEN NE=1:GOTO 1170
1140 COLOR 4:PRINT USING "####>";LCNT;:COLOR 0:PRINT MP$(MPC)
1150 LCNT=LCNT+1
1160 IF MPC<MPMX THEN MPC=MPC+1:GOTO 1120 ELSE PRINT #1,HALT$;:NE=0
1170 FOR IN=1 TO MPC-NE
1180 PRINT #2,MP$(IN)
1190 N=N+1
1200 NEXT
1210 IF NE=0 THEN ERASE MP$:MPF!=FRE(0):DIM MP$(MPMX):PRINT #1,CR$;:GOTO 1110
1220 KEY ON:CLOSE #2:COLOR C2:PRINT :PRINT "> ( COMPLETED !! )"
1230 PRINT #1,CHR$(&H18);CR$;
1240 COLOR C4:
1250 ON ERROR GOTO 480:
1260 RETURN 230
1270 '----- PC --> MELCOM
1280 PRINT "> PC-9800 ---> MELCOM"
1290 INPUT "> PC-9800 drive NO. ";F1$
1300 IF F1$<>" " THEN F1$=F1$+" ":
1310 INPUT "> PC-9800 filename ";F1N$:PRINT
1320 F1$=F1$+F1N$
1330 INPUT "> MELCOM filename ";F2$:PRINT
1340 INPUT "> SURE ( Y/N ) ";YN$
1350 IF YN$<>"Y" THEN 870
1360 OPEN F1$ FOR INPUT AS #2
1370 PRINT #1, CR$;ESC$;"E";
1380 PRINT #1,"D ";CHR$(&H27);F2$;CHR$(&H27)
1390 IF INPUT$(1,1)<>"!" THEN 1390
1400 PRINT #1,"C ME TO ";CHR$(&H27);F2$;CHR$(&H27)
1410 IF INPUT$(1,1)<>"!" THEN 1410
1420 IF INPUT$(1,1)<>". " THEN 1420
1430 LCNT=1
1440 WHILE NOT EOF(2)
1450 LINE INPUT #2,B$
1460 PRINT #1,B$;:COLOR C5:PRINT USING "####<";LCNT;:COLOR C4:PRINT B$
1470 LCNT=LCNT+1
1480 PRINT #1,CR$;
1490 IF INPUT$(1,1)<>". " THEN 1490
1500 WEND
1510 PRINT #1,ESC$;"F";ESC$;"E"
1520 B$=INPUT$(LOC(1),1):B$="":CLOSE 2:KEY ON
1530 COLOR C2:PRINT :PRINT "> ( WORKING !!! )"
1540 PRINT #1,"C ";F2$;" OVER ";F2$;" (LN)"
1550 PRINT #1,"E ";F2$
1560 PRINT #1,"DE1";CR$;"C ";F2$;" OVER ";F2$;" ,1,1"
1570 IF INPUT$(1,1) <> "*" THEN 1570
1580 IF LOC(1)<>0 THEN 1570 ELSE PRINT #1,"E"
1590 IF INPUT$(1,1)<>"!" THEN 1590
1600 PRINT :PRINT "> ( COMPLETED !! )"
1610 D$=INPUT$(LOC(1),#1):PRINT #1,CHR$(&H18);CR$;:COLOR C4
1620 ON ERROR GOTO 480
1630 RETURN 230
1640 REM *****
1650 PRINT CSRCL$;:PRINT #1,HALT$;: '----- HELP
1660 COLOR C4:CLS:GOSUB 2020:
1670 COLOR C4:PRINT #1,CR$;:RETURN 280
1680 PRINT CSRCL$;:PRINT :COLOR C5:'----- FILES
1690 ON ERROR GOTO *ERTRP
1700 INPUT "> FILES n drive No. ";D:IF D=0 THEN D=1
1710 COLOR C4:FILES D:COLOR C5:FC=DSKF(D)
1720 PRINT "> *** ";FC;" clusters free ***":COLOR C4
1740 PRINT #1,CR$;:ON ERROR GOTO 480:RETURN 230
1750 '----- check F

```

```

1760 PRINT CSRCL$;:COLOR C5:PRINT :PRINT "> ( CHECK FILE ( push SPACE key --> SL
OW SPEED DISPLAY ! ) )"
1770 ON ERROR GOTO *ERTRP
1780 COLOR C4:INPUT ">> drive No. ";FC$:IF FC$<>" THEN FC$=FC$+";"
1790 INPUT "> filename ";FCN$:FC$=FC$+FCN$
1800 IF FCN$="" GOTO 1910
1810 IF K2 THEN LPRINT FCN$
1820 OPEN FC$ FOR INPUT AS #2
1830 WHILE NOT EOF(2)
1840 LINE INPUT #2,CF$
1850 PRINT CF$
1860 IF K2=1 THEN LPRINT CF$
1870 IF HEX$(INP(&HE9))="BF" THEN FOR I=1 TO 500:NEXT I
1880 WEND
1890 CLOSE #2:COLOR C5:PRINT :PRINT "> ( END )":BEEP:COLOR C4
1900 IF INKEY$<>" THEN 1900
1910 PRINT #1,CR$;:ON ERROR GOTO 480:RETURN 230
1920 *LPTSET : '----- LPT SET
1930 PRINT:PRINT:PRINT
1940 INPUT " > use PRINTER ( Y/N ) ";A$
1950 IF A$="N" OR A$="" THEN RETURN ELSE IF A$<>"Y" THEN BEEP:PRINT " > CAP
S lock ...";:GOTO 1940
1960 IF INKEY$<>" THEN 1960
1970 PRINT:PRINT:PRINT " >> PRINTER ON !!":PRINT
1980 PRINT " N: high speed pica P: proportional Q: condense"
1990 PRINT " E: elite H: high density pica":PRINT
2000 INPUT " >>> SELECT PRINTER MODE";LPT$:LPRINT ESC$;LPT$;
2010 RETURN
2020 PRINT :PRINT
2030 PRINT "> << function key COMMAND LIST >> HELPキー テ` イツテ`モ` ヒョウシ` シマス。":PRINT
">"
2040 PRINT "> f.1 ..... TERM TERMINALモート`へ。 (shift)+(STOP) テ` BASICモ-
ート`へ。"
2050 PRINT "> f.2 ..... CAN CANCEL シンコ`ウ`ヲ` オクル。1キ`ョウ` マッショウ。"
2060 PRINT "> f.3 ..... NTSS N1ネットワーク`へノ` linkコマント`"
2070 PRINT "> f.4 ..... ID Account(トウロクハ`ンコ`ウ)`ヲ` オクル。"
2080 PRINT "> f.5 ..... HALT HALT シンコ`ウ`ヲ` オクル。(DISPLAY ノ` STOP)"
2090 PRINT "> f.6 ..... TRANS file テンソウ`モ-ト` (メッセ-ジ` ニ` シタカ` ッテ` ソウサスル。)"
2100 PRINT "> f.7 ..... FILES floppydisk ノ` directory`ヲ` ヒョウシ`スル。"
2110 PRINT "> f.8 ..... LPTon/off PRINTER switch ON <---> OFF"
2120 PRINT "> f.9 ..... check F floppydisk ノ` file ノ` ナイヨウ`ヲ` ヒョウシ`スル。"
2130 PRINT "> f.10 ..... END LOGOFF ノ`チ` PROGRAM END (BASICモ-ト`ニ` モ-トル。
)"
2140 PRINT "> HELP ..... HELP イツテ`モ` function key COMMAND LIST`ヲ` ヒョウシ`ス
ル。"
2150 PRINT "> STOP ..... BREAK キ- BREAKシンコ`ウ`ヲ` オクル。 マクハ` file` テンソウ`チュウノ` INTE
RRUPT。"
2160 RETURN
2170 *TITLE
2180 PRINT :PRINT
2190 PRINT " =====
2200 PRINT " ===
2210 PRINT " === KADAI TSS TERMINAL CONTROL PROGRAM FOR PC-9800 ===
2220 PRINT " === 'KTCP219' ( N88-DISK BASIC ) ===
2230 PRINT " === 1985,May,18 Ver. 2.19 by S.Kuroki ===
2240 PRINT " ===
2250 PRINT " =====
2260 PRINT
2270 RETURN
2280 PRINT CSRCL$;:COLOR C5:'----- ID
2290 PRINT " == ACCOUNT SELECT ! == "
2300 PRINT " 1: #####,#####,#####"
2310 PRINT " 2: #####,#####,#####"
2320 PRINT " 3: #####,#####,#####"
2330 PRINT " >> NO. ( 1 - 3 ) ";:INPUT ID:COLOR C4:IF ID<1 OR ID>3 T
HEN BEEP:GOTO 2330
2340 ON ID GOTO 2350,2360,2370

```

```

2350 PRINT #1,"#####,#####,#####":RETURN
2360 PRINT #1,"#####,#####,#####":RETURN
2370 PRINT #1,"#####,#####,#####":RETURN
2380 '=====
2390 *ERTRP
2400 PRINT:PRINT "> *** PC-ERROR code="ERR" , at ("ERL") ***":PRINT
2410 PRINT #1,CR$::RESUME 240
2420 '=====
2430 *INTRPT
2440 CLOSE #2:PRINT:PRINT "> INTERRUPT !":PRINT:BEEP:COLOR 0
2450 RETURN 230
2460 '=====
2470 *KYUDAIF:'***** FILE TRANSFER FOR KYUDAI
2480 ON ERROR GOTO *ERTRP
2490 PRINT "> * KYUDAI * 1: FACOM --> PC-9800 "
2500 PRINT "> 2: PC-9800 --> FACOM"
2510 INPUT "> SELECT ( 1 OR 2 :cancel -> other key ) ";C:PRINT
2520 IF C<1 OR C>2 THEN PRINT #1,CR$::GOTO 2910
2530 INPUT "> SURE ( Y/N ) ";YN$:PRINT
2540 IF YN$<>"Y" THEN 2490
2550 IF C=2 THEN GOTO 2930
2560 '----- FACOM --> PC
2570 PRINT "> FACOM ---> PC-9800"
2580 INPUT "> FACOM 'DSNAME' ";F1$
2590 INPUT "> ... QUALIFIER - ";F1Q$:PRINT
2600 INPUT "> PC-9800 drive NO. ";F2$
2610 IF F2$<>" " THEN F2$=F2$+" ":
2620 INPUT "> PC-9800 filename ";F2N$:PRINT
2630 F2$=F2$+F2N$
2640 INPUT "> SURE ( Y/N ) ";YN$
2650 IF YN$<>"Y" THEN 2490
2660 OPEN F2$ FOR OUTPUT AS #2
2670 TRANS$="LIST "+F1$:IF F1Q$<>" " THEN TRANS$=TRANS$+"."+F1Q$
2680 PRINT:PRINT "> !!!! WAIT A MINUTE !!!!":PRINT:MPF!=FRE(0)
2690 MPT$=INPUT$(LOC(1),1):PRINT #1, TRANS$:CR$::MPT$=""
2700 LINE INPUT #1,MPT$:IF MPT$="" THEN 2700
2710 IF INSTR(MPT$,"KEQ")=0 THEN PRINT MPT$:GOTO 2700
2720 IF INSTR(MPT$,"KEQ528001")<>0 GOTO 2750
2730 PRINT:PRINT"> *** TRANS ERROR *** >>RETRY ON READY MODE !!"
2740 PRINT:BEEP:PRINT MPT$:CLOSE #2:KILL F2$:GOTO 2900
2750 PRINT MPT$:
2760 N=1:LCNT=1
2770 MPC=1:NE=1
2780 LINE INPUT #1,MPT$:IF MPT$="" THEN 2780
2790 IF INSTR(MPT$,"KEQ52802I END OF DATA")<>0 THEN PRINT MPT$:NE=1:GOTO 28
40
2800 MP$(MPC)=MID$(MPT$,6)
2810 COLOR 4:PRINT USING "###>";LCNT::COLOR 0:PRINT MP$(MPC)
2820 LCNT=LCNT+1
2830 IF MPC<MPMX THEN MPC=MPC+1:GOTO 2780 ELSE PRINT #1,HALT$::NE=0
2840 FOR IN=1 TO MPC-NE
2850 PRINT #2,MP$(IN)
2860 N=N+1
2870 NEXT
2880 IF NE=0 THEN ERASE MP$:MPF!=FRE(0):DIM MP$(MPMX):PRINT #1,HALT$::GOTO 2770
2890 PRINT:PRINT "> COMPLETE"
2900 KEY ON:CLOSE #2:PRINT #1,ESC$:CHR$(&H18);
2910 ON ERROR GOTO 480
2920 RETURN 230
2930 '----- PC --> FACOM
2940 PRINT "> PC-9800 ---> FACOM"
2950 INPUT "> PC-9800 drive NO. ";F1$
2960 IF F1$<>" " THEN F1$=F1$+" ":
2970 INPUT "> PC-9800 filename ";F1N$:PRINT
2980 F1$=F1$+F1N$
2990 INPUT "> FACOM 'DSNAME' ";F2$
3000 INPUT "> ... DATA SET TYPE - ";F3$:PRINT

```

```
3010 INPUT ">          SURE ( Y/N )          ";YN$
3020  IF YN$<>"Y" THEN 2490
3030 OPEN F1$ FOR INPUT AS #2
3040 TRANS$="EDIT "+F2$+" NEW "+F3$
3050 PRINT #1,TRANS$;CR$;
3060 LINE INPUT #1,MPT$
3070  IF INSTR(MPT$,"READY")=0 GOTO 3100
3080  PRINT:PRINT "> *** COMMAND ERROR ***>>RETRY !":PRINT
3090  BEEP:PRINT MPT$:GOTO 3290
3100  IF INSTR(MPT$,"INPUT")=0 THEN PRINT MPT$:GOTO 3060 ELSE PRINT MPT$
3110 B$=INPUT$(1,1):IF B$<>" " THEN GOTO 3110
3120 PRINT #1,ESC$;"E";
3130 LCNT=1
3140 WHILE NOT EOF(2)
3150  LINE INPUT #2,B$:IF B$="" THEN 3150
3160  PRINT #1,B$;CR$;
3170  COLOR C5:PRINT USING "####<";LCNT;:COLOR C4:PRINT B$
3180  B$=INPUT$(1,1):IF B$<>CR$ GOTO 3180
3190  B$=INPUT$(1,1):IF B$<>" " GOTO 3190
3200  LCNT=LCNT+1
3210 WEND
3220 PRINT #1,"";CR$; ESC$;"E";
3230 PRINT "> COMPLETE"
3240 LINE INPUT #1,MPT$:IF MPT$="" THEN 3240
3250  IF INSTR(MPT$,"EDIT")=0 THEN PRINT MPT$:GOTO 3240 ELSE PRINT MPT$
3260 L=LOC(1):IF L<>0 THEN D$=INPUT$(L,1):GOTO 3260
3270  PRINT #1,"END SAVE";CR$;CHR$(&H18);
3280 IF INKEY$<>" " THEN 3280
3290 CLOSE #2:KEY ON
3300 ON ERROR GOTO 480
3310 RETURN 230
```